# Maintainer: Evangelos Foutras <evangelos@foutrelis.com>
# Contributor: Andreas Radke <andyrtr@archlinux.org>
# Contributor: Tobias Kieslich <tobias funnychar archlinux.org>

pkgname=xfce4-dev-tools
pkgver=4.16.0
pkgrel=1
pkgdesc="Xfce developer tools"
arch=('x86_64')
url="https://www.xfce.org/"
license=('GPL2')
depends=('glib2' 'intltool' 'gtk-doc')
optdepends=('docker: xfce-build helper script')
source=(https://archive.xfce.org/src/xfce/xfce4-dev-tools/${pkgver%.*}/$pkgname-$pkgver.tar.bz2)
sha256sums=('f50b3070e66f3ebdf331744dd1ec5e1af5de333965d491e15ce05545e8eb4f04')

build() {
  cd "$srcdir/$pkgname-$pkgver"

  ./configure \
    --prefix=/usr \
    --sysconfdir=/etc \
    --libexecdir=/usr/lib \
    --localstatedir=/var
  make
}

package() {
  cd "$srcdir/$pkgname-$pkgver"
  make DESTDIR="$pkgdir" install
}

# vim:set ts=2 sw=2 et:
